Gröbner Bases

Victor Adamchik

Carnegie Mellon University

Lagrange Multipliers

We are to solve constrained optimization problems. In linear case the problem is a linear programming one and can be solved using the simplex algorithm. In order to gain some intuition, let us cosider the simple case

$$\textit{minimize } f(x,\ y)$$

$$\textit{subject to } h(x,\ y) = 0$$

Differentiates $h(x,\ y) = 0$, wrt x

$$\frac{\partial h}{\partial x}\frac{dx}{dx} + \frac{\partial h}{\partial y}\frac{dy}{dx} = 0$$

Using the notation for the tangent $T = \left\{1,\ \frac{dy}{dx}\right\}$ and the gradient (or normal vector) $\nabla \mathrm{h} = \left\{\frac{dh}{dx},\ \frac{dh}{dy}\right\}$, the above equation can be rewritten as a scalar product

$$T \,.\, \nabla \mathrm{h} = 0$$

The tangent of the curve is normal to the gradient all the time.





At an extremum of f, we must have $T \,.\, \nabla \mathrm{f} = 0$, so f and h are tangent to each other, Thus, T is orthogonal to both gradients $\nabla \mathrm{f}$ and ∇h at an extrema, and therefore $\nabla \mathrm{f}$ and $\nabla \mathrm{h}$ must be parallel.

$$\nabla \mathrm{f} + \lambda \nabla \mathrm{h} = 0$$

The original problem is transfered into

$$h(x,\ y) = 0$$

$$\nabla f + \lambda \nabla h = 0$$

that can be now solved.

It is convenient to introduce the Lagrangian associated with the constrained problem, defined as

$$L(x, y) = f(x, y) + \lambda g(x, y)$$

The new variable λ is called Lagrang multiplier. We solve $\nabla_{x,y,\lambda} L(x, y) = 0$ to find the constrained extrema. Note, that $\nabla_\lambda L(x, y) = 0$, implies $g(x, y) = 0$.

■ **Example.**

$$x y \to \max$$

$$x^2 + 4 y^2 = 8$$

We define the Lagrangian

$$L(x, y) = x y + \lambda \left(x^2 + 4 y^2 - 8\right)$$

Compute derivatives

```
D[x y + λ (x^2 + 4 y^2 - 8), #] & /@ {x, y, λ}
```

```
{y + 2 x λ, x + 8 y λ, -8 + x^2 + 4 y^2}
```

Combining first two equations, we get

$$y + 2 \lambda x = y - 2 \lambda 8 y \lambda = y\left(1 - 16 \lambda^2\right) = 0$$

$$\lambda = \pm \frac{1}{4}$$

Then

$$x = -8 y \lambda \to x = \pm 2 y$$

$$-8 + x^2 + 4 y^2 = -8 + 4 y^2 + 4 y^2 = -8 + 8 y^2$$

$$y = \pm 1$$
$$x = \pm 2$$

See the following tutorial http://www.slimy.com/~steuard/teaching/tutorials/Lagrange.html

Concentric Circles

Given three concentric circles (they have their centers at the same point) of radii r_1, r_2, and r_3. Calculate the maximal area of the triangle that has one vertex at each of the circles. For $r_1 = 1$, $r_2 = 2$, $r_3 = 3$ calculate the explicit value of the area.



■ Solution

Let p1, p2, and p3 be the vertices of the triangle:

```
p1 = {r1, 0};
p2 = {x2, y2};
p3 = {x3, y3};
```

Without loss of generality we can assume the point p1 to lie on the x-axes, and p2 and p3 are lying on other two circles. This gives us two equations

$x2^2 + y2^2 = r2^2$
$x3^2 + y3^2 = r3^2$

We compute the area of the triangle by a general formula for a polygon, given by

$$A = \frac{1}{2}\left|\begin{matrix} x_1, x_2 \\ y_1, y_2 \end{matrix}\right| + \frac{1}{2}\left|\begin{matrix} x_2, x_3 \\ y_2, y_3 \end{matrix}\right| + \ldots + \frac{1}{2}\left|\begin{matrix} x_n, x_1 \\ y_n, y_1 \end{matrix}\right|$$

Hence,

```
A = 1/2 Det[{p1, p3}] + 1/2 Det[{p3, p2}] + 1/2 Det[{p2, p1}] // Expand
```

$$-\frac{\text{r1 y2}}{2} + \frac{\text{x3 y2}}{2} + \frac{\text{r1 y3}}{2} - \frac{\text{x2 y3}}{2}$$

In order to find the max area we will be using Lagrange multipliers. In our problem we need to maximize the function A

$$A = -\frac{\text{r1 y2}}{2} + \frac{\text{r1 y3}}{2} - \frac{\text{x2 y3}}{2} + \frac{\text{x3 y2}}{2}$$

subject to constraints

$$x2^2 + y2^2 - r2^2 = 0$$
$$x3^2 + y3^2 - r3^2 = 0$$

The above system of three equations can be rewritten by defining the Lagrangian *L* and two Lagrang multipliers:

```
L = A + μ (x2^2 + y2^2 - r2^2)  + λ (x3^2 + y3^2 - r3^2)
```

$$-\frac{r1\ y2}{2} + \frac{x3\ y2}{2} + \frac{r1\ y3}{2} - \frac{x2\ y3}{2} + \left(-r3^2 + x3^2 + y3^2\right)\lambda + \left(-r2^2 + x2^2 + y2^2\right)\mu$$

Finding derivatives wrt to all 6 variables

```
D[L, #] & /@ {x2, y2, x3, y3, λ, μ}
```

$$\left\{-\frac{y3}{2} + 2\ x2\ \mu,\ -\frac{r1}{2} + \frac{x3}{2} + 2\ y2\ \mu,\ \frac{y2}{2} + 2\ x3\ \lambda,\ \frac{r1}{2} - \frac{x2}{2} + 2\ y3\ \lambda,\ -r3^2 + x3^2 + y3^2,\ -r2^2 + x2^2 + y2^2\right\}$$

produce the new system of equation that we solve by the Groebner bases technique.

```
Clear[A];
gb =
 GroebnerBasis[{A - (-r1 y2/2 + x3 y2/2 + r1 y3/2 - x2 y3/2), -y3/2 + 2 x2 μ,
    -r1/2 + x3/2 + 2 y2 μ, y2/2 + 2 x3 λ, r1/2 - x2/2 + 2 y3 λ, -r3^2 + x3^2 + y3^2,
    -r2^2 + x2^2 + y2^2},
      {A, r1, r2, r3}, {x2, y2, λ, μ, x3, y3},
      MonomialOrder -> EliminationOrder][[1]]
```

$$256 A^6 + 16 A^4 r1^4 - 160 A^4 r1^2 r2^2 - 8 A^2 r1^6 r2^2 + 16 A^4 r2^4 + 32 A^2 r1^4 r2^4 + r1^8 r2^4 - 8 A^2 r1^2 r2^6 - 2 r1^6 r2^6 + r1^4 r2^8 - 160 A^4 r1^2 r3^2 - 8 A^2 r1^6 r3^2 - 160 A^4 r2^2 r3^2 - 16 A^2 r1^4 r2^2 r3^2 - 2 r1^8 r2^2 r3^2 - 16 A^2 r1^2 r2^4 r3^2 + 2 r1^6 r2^4 r3^2 - 8 A^2 r2^6 r3^2 + 2 r1^4 r2^6 r3^2 - 2 r1^2 r2^8 r3^2 + 16 A^4 r3^4 + 32 A^2 r1^4 r3^4 + r1^8 r3^4 - 16 A^2 r1^2 r2^2 r3^4 + 2 r1^6 r2^2 r3^4 + 32 A^2 r2^4 r3^4 - 6 r1^4 r2^4 r3^4 + 2 r1^2 r2^6 r3^4 + r2^8 r3^4 - 8 A^2 r1^2 r3^6 - 2 r1^6 r3^6 - 8 A^2 r2^2 r3^6 + 2 r1^4 r2^2 r3^6 + 2 r1^2 r2^4 r3^6 - 2 r2^6 r3^6 + r1^4 r3^8 - 2 r1^2 r2^2 r3^8 + r2^4 r3^8$$

This equation represents the maximal area of the triangle *A* of radii r_1, r_2, and r_3. Lastly, we compute that area for $r_1 = 1$, $r_2 = 2$, $r_3 = 3$:

```
gb /. {r1 → 1, r2 → 2, r3 → 3}
```

$$14400 + 2128 A^2 - 6272 A^4 + 256 A^6$$

```
NSolve[% == 0, A]
```

```
{{A → -4.90482}, {A → -1.32906}, {A → 0. - 1.15052 i},
 {A → 0. + 1.15052 i}, {A → 1.32906}, {A → 4.90482}}
```

It follows, the max area is 4.90482.